

## Table de la loi du groupe symétrique S3

Création du groupe symétrique S3

```
S3:=Dom::SymmetricGroup(3);  
Dom::SymmetricGroup(3)
```

(ceci "crée" une "instance" S3 de "la classe" SymmetricGroup, sur laquelle on pourra exercer les "méthodes" allElements, size, \_mult, etc. , voir ci-après)

Pour consulter l'aide: lancer le "Help Browser", puis menus: "Contents", "Domains", "SymmetricGroup"

(remarquer les 2 m à symétrique pour les anglais)

Liste des éléments de S3

```
les_sigma:=S3::allElements();  
[[2, 3, 1], [3, 2, 1], [1, 3, 2], [3, 1, 2], [1, 2, 3], [2, 1, 3]]
```

(remarquer la façon dont MuPAD représente les permutations)

L'ordre du groupe S3

```
ord:=S3::size();  
6
```

(ouf !)

Pour nommer les éléments de S3 comme on a l'habitude ...

```
Noms:=["e", "t1", "t2", "t3", "c", "c^2"];  
indice_de_mes_sigma:=[5, 3, 2, 6, 1, 4];  
mes_sigma:=[les_sigma[indice_de_mes_sigma[k]] $k=1..ord];  
["e", "t1", "t2", "t3", "c", "c^2"]  
[5, 3, 2, 6, 1, 4]  
[[1, 2, 3], [1, 3, 2], [3, 2, 1], [2, 1, 3], [2, 3, 1], [3, 1, 2]]
```

Calcul de la table de la loi (dans l'ordre que j'ai choisi !)

```
table_de_S3:= array(1..ord,1..ord,  
[ [ S3::_mult( mes_sigma[i],  
mes_sigma[j])  
$j=1..ord] $i=1..ord]);  

$$\begin{pmatrix} [1, 2, 3] & [1, 3, 2] & [3, 2, 1] & [2, 1, 3] & [2, 3, 1] & [3, 1, 2] \\ [1, 3, 2] & [1, 2, 3] & [2, 3, 1] & [3, 1, 2] & [3, 2, 1] & [2, 1, 3] \\ [3, 2, 1] & [3, 1, 2] & [1, 2, 3] & [2, 3, 1] & [2, 1, 3] & [1, 3, 2] \\ [2, 1, 3] & [2, 3, 1] & [3, 1, 2] & [1, 2, 3] & [1, 3, 2] & [3, 2, 1] \\ [2, 3, 1] & [2, 1, 3] & [1, 3, 2] & [3, 2, 1] & [3, 1, 2] & [1, 2, 3] \\ [3, 1, 2] & [3, 2, 1] & [2, 1, 3] & [1, 3, 2] & [1, 2, 3] & [2, 3, 1] \end{pmatrix}$$

```

(consulter l'aide de array et celle de Dom::SymmetricGroup\_mult pour s'assurer que la composition est effectuée selon l'ordre désiré)

### La même table, avec les beaux noms !

Pour chaque case de la table, on récupère l'indice (dans la liste "mes\_sigma") de la permutation correspondante.

Puis on écrit la table avec les noms correspondants (text2expr permet de lever les " ").

```
table_de_S3bis:= array(1..ord,1..ord,  
[ [ contains(mes_sigma,table_de_S3[i,j])  
    $j=1..ord] $i=1..ord]);  
belle_table_de_S3:=map(table_de_S3bis, x->text2expr(Noms[x]))
```

$$\begin{pmatrix} 1 & 2 & 3 & 4 & 5 & 6 \\ 2 & 1 & 5 & 6 & 3 & 4 \\ 3 & 6 & 1 & 5 & 4 & 2 \\ 4 & 5 & 6 & 1 & 2 & 3 \\ 5 & 4 & 2 & 3 & 6 & 1 \\ 6 & 3 & 4 & 2 & 1 & 5 \end{pmatrix}$$

$$\begin{pmatrix} e & t1 & t2 & t3 & c & c^2 \\ t1 & e & c & c^2 & t2 & t3 \\ t2 & c^2 & e & c & t3 & t1 \\ t3 & c & c^2 & e & t1 & t2 \\ c & t3 & t1 & t2 & c^2 & e \\ c^2 & t2 & t3 & t1 & e & c \end{pmatrix}$$